

Equation Listing

Column	Name	Formula	Description
U	distribution conduit inv	=IF(wireless_cap_ind=0,calculations!AU2,0)	reports total investment in distribution conduit materials
V	distribution conduit placement inv	=IF(wireless_cap_ind=0,calculations!AV2,0)	reports total investment in distribution conduit placement
W	distribution poles inv	=IF(wireless_cap_ind=0,calculations!AS2,0)	reports total investment in distribution poles and pole setting
X	distribution buried placement inv	=IF(wireless_cap_ind=0,calculations!AT2,0)	reports total investment in distribution buried cable placement
Y	rock plcmnt mult	=IF(calculations!P2=0,rock_mult,0)	reports rock placement multiplier to be used by feeder module in modifying placement investment as may be required by shallow bedrock conditions
Z	difficult surface mult	=IF(calculations!P2=0,diff_sfc,0)	reports difficult surface placement multiplier to be used by feeder module in modifying placement investment as may be required by difficult surface conditions
AA	water table depth, ft	=IF(calculations!P2=0,'cluster input data'!V2,0)	reports water table depth for use by feeder module in modifying manhole investment in presence of high water table
AB	effective distribution fill	=IF(AND(wireless_cap_ind=0,calculations!P2=0),(0.5*calculations!R2/(calculations!S2*calculations!U2)/(calculations!AA2+calculations!AB2*max_cable)),0)	reports achieved distribution cable fill factor as computed at distribution side of SAI
AC	number of high-density RTs	=IF(wireless_cap_ind=0,IF(calculations!F2=1,calculations!CA2,0),0)	reports total high-density remote terminals required in cluster; includes effects of branch or backbone subdivision, if required
AD	high-density RT investment	=IF(wireless_cap_ind=0,IF(calculations!F2=1,calculations!CB2,0),IF(calculations!P2=0,IF(clstr_tot_lines<=brdcast_thresh,clstr_tot_lines*wireless_cap,brdcast_common_inv*CEILING(clstr_tot_lines/brdcast_lines_max,1)+brdcast_var_inv*clstr_tot_lines,0)))	reports total investment in high-density DLC remote terminals for cluster; computes representative wireless investment totals if wireless "cap" calculations are enabled and local distribution investment exceeds either point-to-point or broadcast "cap"
AE	number of low-density DLC RTs	=IF(wireless_cap_ind=0,IF(calculations!F2=1,calculations!BY2,0),0)	reports total low-density remote terminals required in cluster; includes effects of branch or backbone subdivision, if required
AF	low-density DLC and T1 road terminal and repeater investment	=IF(wireless_cap_ind=0,IF(calculations!F2=1,calculations!BZ2,0)+calculations!BP2*calculations!AK2+calculations!BS2,0)	reports investment in low-density DLC remote terminals and in T1 equipment for outliers

Equation Listing

Column	Name	Formula	Description
AG	fiber strands required	=IF(wireless_cap_ind=0,IF(calculations!P2=1,calculations!CC2,0),IF(calculations!P2=0,4,0))	reports total fiber strands needed by RTs in cluster; if wireless cap exceeded, reports four fibers for radio equipment
AH	SAI investment	=IF(wireless_cap_ind=0,calculations!BD2+calculations!BG2,0)	reports total SAI investment for cluster, including indoor SAI totals when high-rise calculations are invoked
AI	terminal investment	=IF(wireless_cap_ind=0,calculations!DT2,0)	reports total investment in terminals (interface between drops and distribution cable)
AJ	drop investment	=IF(wireless_cap_ind=0,calculations!DV2,0)	reports total investment in installed subscriber drops
AK	NID investment	=IF(wireless_cap_ind=0,calculations!DX2,0)	reports total investment in network interface devices
AL	number of DLC lines	=IF(AG2=0,0,calculations!DZ2)	reports total number of lines served by DLC in cluster and outliers
AM	vertical connecting cable length, ft	=IF(wireless_cap_ind=0,calculations!AW2,0)	reports vertical (optical fiber) connecting cable to feeder module for sizing and investment computation when backbone cable is subdivided
AN	horizontal connecting cable length, ft	=IF(wireless_cap_ind=0,calculations!AX2,0)	reports horizontal (optical fiber) connecting cable to feeder module for sizing and investment computation when branch cables are subdivided
AO	vertical connecting structure distance, ft	=IF(wireless_cap_ind=0,calculations!AY2,0)	reports structure distance for vertical connecting cable, when present
AP	horizontal connecting structure distance, ft	=IF(wireless_cap_ind=0,calculations!AZ2,0)	reports structure distance for horizontal connecting cable, when present
AQ	average loop length in cluster, ft	=IF(wireless_cap_ind=0,calculations!BT2,0)	reports average loop length within cluster
AR	maximum loop length, ft	=IF(wireless_cap_ind=0,calculations!BV2,0)	reports maximum loop length in cluster
AS	cluster ID	='cluster input data'!E2	repeats cluster ID from cluster input sheet
AT	cluster serial number	=calculations!CG2	reports basic cluster serial number without outlier extensions for use by interface in totalling cluster investments
AU	wireless cap indicator	=IF(calculations!P2=0,wireless_cap_ind,0)	reports whether wireless cap is reached for cluster; saved in workfile
AV	lines affected by wireless cap	=IF(AND(AU2=1,calculations!P2=0),calculations!DZ2,0)	If wireless cap applies, records total lines in cluster and outliers; save in workfile
AW	cable+structure+DLC inv less wireless estimate	=IF(AND(AU2=1,calculations!P2=0),calculations!EA2-output!AD2,0)	calculates investment saved by employment of wireless assumptions

Equation Listing

Column	Name	Formula	Description
AX	calc buried fraction -- main cluster	=IF(AND(\$AU2=0,calculations!\$P2=0),calculations!CD2,0)	reports calculated buried fraction
AY	calc aerial fraction -- main cluster	=IF(AND(\$AU2=0,calculations!\$P2=0),calculations!CE2,0)	reports calculated aerial structure fraction
AZ	calc u/g fraction -- main cluster	=IF(AND(\$AU2=0,calculations!\$P2=0),calculations!CF2,0)	reports underground structure fraction
BA	overall wtd avg loop length	=IF(calculations!P2=0,(AQ2*lines_adj+('cluster input data'!AQ2*IF(diff_sfc>1,dstnc_mult,1)+calculations!EG2+F2+G2)*'cluster input data'!AR2)/('cluster input data'!AR2+lines_adj),0)	computes overall weighted average loop length, using average loop length in cluster and input average outlier length, weighted by lines
BB	cluster fraction of wire center lines	=IF(calculations!P2=0,'cluster input data'!AP2,0)	computes wtd average adjustment for cluster

Equation Listing

Column	Name	Formula	Description
A	Company	The data for this sheet is taken from output sheet of the distribution module.	
B	operating company indicator		
C	wire center		
D	CBG geocode		
E	quadrant		
F	main feeder distance		
G	subfeeder		
H	distribution distance		
I	total lines		
J	density, lines/sq mi		
K	area, sq mi		
L	business lines		
M	residential lines		
N	SA lines		
O	public lines		
P	households		
Q	single-line business lines		
R	distribution cable, underground		
S	distribution cable, buried		
T	distribution cable, aerial		
U	distribution conduit		
V	distribution conduit placement		
W	distribution poles		
X	distribution buried plcmnt		
Y	rock plcmnt mult		
Z	difficult surface mult		
AA	water table depth, ft		
AB	effective distribution cable fill		
AC	number of TR-303 RTs		
AD	TR-303 investment		

Equation Listing

Column	Name	Formula	Description
AE	number of low-density DLC RTs		
AF	low-density DLC investment		
AG	fiber strands required		
AH	SAI inv		
AI	terminal inv		
AJ	drop inv		
AK	NID inv		
AL	number of DLC lines		
AM	vertical connecting cable length, ft		
AN	horizontal connecting cable length, ft		
AO	vertical structure distance, ft		
AP	horizontal structure distance, ft		
AQ	average loop length, ft		
AR	maximum loop length, ft		
AS	cluster ID		
AT	cluster serial number		
AU	wireless cap indicator		
AV	lines affected by wireless cap		
AW	cable+structure+DLC inv less wireless estimate		
AX	calc buried fraction -- main cluster		
AY	calc aerial fraction -- main cluster		
AZ	calc u/g fraction -- main cluster		
BA	overall wtd avg loop length		
BB	cluster fraction of wire center lines		

Equation Listing

Column	Name	Formula	Description
A	wire center	=distribution input!C2	Wire center serving this cluster
B	Quadrant	=distribution input!E2	Quadrant where this cluster is located.
C	main feeder distance, ft	=distribution input!F2	Main feeder distance from the wire center to the subfeeder splice for this cluster
D	subfeeder distance	=distribution input!G2	Subfeeder distance from the main feeder splice to the center of this cluster
E	basic lines	=distribution input!I2	Total lines in cluster
F	line density per sq mi	=distribution input!J2	Repeats line density in lincs/sq mi for cluster
G	strands required	=distribution input!AG2	Number of fiber strands required to serve the DLC remote terminal(s)
H	copper subfeeder size	=IF(OR(subfeeder_dist=0,basic_segment_type="DLC"),"N/A",AJ2)	repeats assigned copper subfeeder cable cross section, if equipped
I	current segment inv	=IF(AG2=0,0,X2*(AH2*max_copper_inv+VLOOKUP(AG2,copper_inv,2,FALSE)*((1-wtg_copper_const)+wtg_copper_const*labor_adj)))	Investment in copper feeder cable required for feeder segment serving cluster
J	max copper cable inv (subfeeder)	=IF(basic_segment_type<>"Cable",0,max_copper_inv*(subfeeder_dist*AK2)*((1-wtg_copper_const)+wtg_copper_const*labor_adj))	calculates investment in maximum (overflow) copper subfeeder cables, if present
K	fiber subfeeder size	=IF(AND(subfeeder_dist>0,basic_segment_type="DLC"),MAX(min_fiber_subfdr,INDEX(fiber_range,MATCH('distribution input'!\$AG2,VLOOKUP(line_density,density_inputs,3),fiber_range,-1),1)),"N/A")	assigns cable size for fiber subfeeder cable
L	fiber subfeeder investment	=IF(K2="N/A",0,(subfeeder_dist+'distribution input'!AN2+'distribution input'!AM2)*VLOOKUP(K2,fiber_inv,2,FALSE))	calculates fiber subfeeder cable investment
M	fiber main segment investment	=IF(AC2=0,0,X2*VLOOKUP(AL2,fiber_inv,2,FALSE)+AM2*max_fiber_inv*X2)	calculates investment in fiber cable for feeder segment serving cluster, including investment in any overflow cables
N	fiber subfeeder segment investment	=IF(basic_segment_type="DLC",L2,0)	Repeats investment in fiber subfeeder cable
O	segment number	=IF(this_wire_cntr=0,0,IF(AND(this_wire_cntr=next_wire_cntr,this_quadrant=next_quadrant),O3+1,1))	Numbers clusters along feeder route beginning at 1 for cluster nearest wire center
P	segment investment per line	=IF(AD2=0,0,I2/AD2)	Divides segment investment for copper cable by number of lines in current cluster and sum of lines in all clusters served by copper more distant from the wire center
Q	cumulative investment per line	=IF(OR(O1=1,ISTEXT(Q1)),P2,P2+Q1)	Accumulates investment per line in copper cable from wire center end of feeder route

Equation Listing

Column	Name	Formula	Description
R	assigned copper cable investment	=IF(basic_segment_type="Cable",basic_lines*Q2,0)	assigns copper cable investment to cluster by multiplying segment investment per line by number of lines in cluster
S	segment investment per strand	=IF(AC2=0,0,M2/AC2)	calculates investment per strand of fiber cable for segment serving this cluster
T	cumulative investment per strand	=IF(OR(O1=1,ISTEXT(T1)),S2,S2+T1)	Accumulates investment per strand along feeder route, beginning at wire center end of route
U	assigned fiber investment	=G2*T2	Assigns share of fiber cable investment to cluster
V	max distance	=IF(OR(W3=1,ISBLANK(W3)),C2,V3)	Computes and repeats maximum distance along feeder route
W	cluster sequence number	=IF(OR(this_quadrant<>B1,this_wire_cntr<>A1),1,W1+1)	Numbers cluster appearance along feeder route, beginning at wire center
X	segment distance	=IF(AND(this_wire_cntr=A1,this_quadrant=B1),C2-C1,C2)	Computes length of main feeder segment connecting previous cluster to current cluster
Y	basic segment type	=IF(this_wire_cntr="", "", IF(G2=0,"Cable","DLC"))	Indicates whether segment is served by fiber feeder ("DLC") or copper feeder ("Cable")
Z	secondary segment type	=IF(this_quadrant=0,"",IF(OR(W3=1,AND(this_wire_cntr=next_wire_cntr,this_quadrant=next_quadrant,basic_segment_type=Y3,ISBLANK(Z3))), "",IF(basic_segment_type<>Y3,Y3,Z3)))	Indicates whether clusters farther along feeder route from current cluster use feeder cable type different from that used by current cluster
AA	cumulative required copper pairs	=IF(basic_segment_type="Cable",IF(OR(W3=1,AND(W3<>1,basic_segment_type<>Y3,next_quadrant<>this_quadrant,Z2<>"Cable")),ISBLANK(AA3)),basic_lines/VLOOKUP(line_density,density_inputs,2),basic_lines/VLOOKUP(line_density,density_inputs,2)+AA3),IF(W3=1,0,AA3))	Accumulates required copper cable pair count, including cable sizing factor, from far end of cable toward wire center
AB	cumulative lines served by fiber	=IF(basic_segment_type="DLC",IF(OR(W3=1,AND(W3<>1,basic_segment_type<>Y3,Z2<>"DLC")),ISBLANK(AB3)),basic_lines,basic_lines+AB3),IF(Z2="DLC",AB3,0))	accumulates lines served by fiber from remote end of feeder route
AC	total number of fiber strands	=IF(AB2=0,0,IF(W3=1,G2,IF(AB2=AB3,AC3,G2+AC3)))	accumulates number of fiber strands required from remote end of feeder route
AD	cumulative copper lines	=IF(basic_segment_type="Cable",IF(OR(W3=1,AND(W3<>1,basic_segment_type<>Y3,next_quadrant<>this_quadrant,Z2<>"Cable")),ISBLANK(AA3)),basic_lines,basic_lines+AD3),IF(W3=1,0,AD3))	accumulates lines served by copper from remote end of feeder route
AE	cumulative fiber lines	=IF(basic_segment_type="DLC",IF(OR(W3=1,AND(W3<>1,basic_segment_type<>Y3,next_quadrant<>this_quadrant,Z2<>"DLC")),ISBLANK(AA3)),basic_lines,basic_lines+AE3),IF(W3=1,0,AE3))	Computes cumulative fiber line count from far end of feeder route for use in feeder conduit calculation
AF	calc copper feeder fill	=IF(OR(AG2=0,basic_segment_type="DLC"),0,AD2/(AG2+AH2*max_copper))	calculates achieved copper feeder fill factor for segment serving cluster

Equation Listing

Column	Name	Formula	Description
AG	equipped feeder cable	=IF(AA2<>0,INDEX(copper_range,MATCH(AA2-(max_copper*AH2),copper_range,-1),1),0)	determines feeder cable size in segment serving cluster
AH	number of maximum-sized main feeder cables	=TRUNC(IF(AA2>max_copper,AA2/max_copper,0))	computes number of "overflow," or maximum-sized, main feeder cables required, if any
AI	sub cable pairs	=IF(basic_segment_type="Cable",basic_lines/VLOOKUP(line_density,density_inputs,2),0)	determines number of copper pairs required in subfeeder cable to serve cluster; includes effect of cable sizing factor
AJ	sub cable size	=IF(AI2<>0,INDEX(copper_range,MATCH(AI2-(max_copper*AK2),copper_range,-1),1),0)	Assigns subfeeder cable size
AK	max sub cables	=TRUNC(IF(AI2>max_copper,AI2/max_copper,0))	Calculates number of overflow subfeeder copper cables, if any
AL	fiber cable size	=IF(AC2=0,0,INDEX(fiber_range,MATCH(AC2/VLOOKUP(line_density,density_inputs,3)-max_fiber*TRUNC(AC2/(max_fiber*VLOOKUP(line_density,density_inputs,3))),fiber_range,-1),1))	assigns fiber cable cross section for segment serving cluster.
AM	max fiber cables	=IF(AC2/VLOOKUP(line_density,density_inputs,3)>max_fiber,TRUNC(AC2/(max_fiber*VLOOKUP(line_density,density_inputs,3))),0)	calculates number of overflow fiber cables main feeder segment serving this cluster, if required
AN	fiber segment cumulative distance	=IF(AND(basic_segment_type<>"DLC",Z2<>"DLC"),0,IF(AND(this_wire_cntr=A1,this_quadrant=B1),X2+AN1,X2))	accumulates fiber segment distance beginning at wire center end of feeder route
AO	fiber segment total distance from end	=IF(OR(basic_segment_type="DLC",Z2="DLC"),IF(OR(W3=1,AND(Y3<>"DLC",Z3<>"DLC")),AN2,AO3),0)	Calculates distance from cluster to end of fiber route
AP	copper feeder conduit inv	=IF(OR(basic_segment_type="Cable",Z2="Cable"),(X2*(1+AH2+spare_tubes_sect))*BU2*conduit_mat_inv_ft,0)	Computes investment in conduit for copper feeder cable for segment, including spare tubes and tubes for overflow cables, if any
AQ	copper feeder manholes	=IF(OR(basic_segment_type="Cable",Z2="Cable"),X2*BU2/VLOOKUP(line_density,density_inputs,7)*(VLOOKUP(line_density,density_inputs,17)+IF('distribution input'!AA2<=water_thresh,(min_water_factor+(max_water_factor-min_water_factor)/water_thresh*(water_thresh-'distribution input'!AA2))*VLOOKUP(line_density,inputs!\$B\$16:\$G\$24,6),0)),0)	Computes investment in manholes for copper feeder cable; includes extra investment for dewatering, which varies linearly with water table depth
AR	copper feeder u/g trenching	=IF(basic_segment_type="Cable",X2*BU2*VLOOKUP(line_density,density_inputs,9)*((1-wtg_excav)+wtg_excav*labor_adj)*('distribution input'!Y2+'distribution input'!Z2-1),0)	Computes investment in underground placement for segment
AS	copper feeder buried placement	=IF(basic_segment_type="Cable",X2*BS2*VLOOKUP(line_density,density_inputs,8)*((1-wtg_excav)+wtg_excav*labor_adj)*('distribution input'!Y2+'distribution input'!Z2-1),0)	Calculates investment in buried feeder placement for segment

Equation Listing

Column	Name	Formula	Description
AT	copper feeder pole inv	=IF(basic_segment_type="Cable", (1+CEILING(X2*BT2/VLOOKUP(line_density, density_inputs,19),1))*(pole_materials+pole_labor*((1-wtg_pole_set)+wtg_pole_set*labor_adj)))*('distribution input'!Y2+'distribution input'!Z2-1),0)	Computes investment in feeder poles for copper cable; poles carry both copper and optical fiber cable when required
AU	fiber feeder conduit inv	=IF(AND(basic_segment_type="Cable", Z2<>"DLC"),0,IF(AO2=0,0,1/AO2*V2*X2*BX2*((1+spare_tubes_sect)*(conduit_mat_inv_ft)+AM2*inner_duct_inv_ft)))	Computes investment in conduit for underground fiber feeder cable, including any required inner duct for overflow cable
AV	fiber fdr pullboxes	=IF(basic_segment_type<>"Cable",IF(AQ2<0,0,X2*BX2/VLOOKUP(line_density,density_inputs,13)*VLOOKUP(line_density,density_inputs,18)),0)	Calculates investment for pullboxes for fiber feeder cable in segment
AW	fiber feeder u/g trenching	=IF(basic_segment_type<>"Cable",X2*BX2*VLOOKUP(line_density,density_inputs,15)*((1-wtg_excav)+wtg_excav*labor_adj))*('distribution input'!Y2+'distribution input'!Z2-1),0)	Computes investment in placement for underground fiber cable for segment
AX	fiber feeder buried placement	=IF(basic_segment_type<>"Cable",X2*BV2*VLOOKUP(line_density,density_inputs,14)*((1-wtg_excav)+wtg_excav*labor_adj))*('distribution input'!Y2+'distribution input'!Z2-1),0)	Calculates investment in buried placement for fiber feeder cable for this segment
AY	fiber feeder poles	=IF(basic_segment_type<>"Cable", (1+CEILING((X2)*BW2/VLOOKUP(line_density,density_inputs,19),1))*(pole_materials+pole_labor*((1-wtg_pole_set)+wtg_pole_set*labor_adj)))*('distribution input'!Y2+'distribution input'!Z2-1),0)	Computes investment in feeder poles for fiber cable for this segment
AZ	cumulative number of lines	=IF(O2=1,basic_lines,basic_lines+AZ3)	Accumulates total lines from remote end of feeder route.
BA	copper feeder conduit segment inv per line	=IF(AD2=0,0,AP2/AD2)	Computes copper conduit investment for the current segment per line
BB	cumulative conduit inv per line	=IF(OR(O1=1,ISTEXT(BB1)),BA2,BA2+BB1)	accumulates per-line conduit investment from beginning of feeder route
BC	assigned copper feeder conduit inv	=IF(basic_segment_type="Cable",basic_lines*BB2+subfeeder_dist*(1+AK2+spare_tubes_sect)*BU2*conduit_mat_inv_ft,0)	Computes cluster's share of copper conduit investment plus conduit required for copper subfeeder. Note that connecting cable and structure distances do not affect this calculation because they are only used with fiber feeder
BD	total feeder manhole sgmt inv per line	=IF(AZ2=0,0,(AQ2+AV2)/AZ2)	computes manhole and pullbox investment per cumulative line
BE	cumulative inv per line	=IF(OR(O1=1,ISTEXT(BE1)),BD2,BD2+BE1)	accumulates per-line investment in manholes and pullboxes from beginning of route

Equation Listing

Column	Name	Formula	Description
BF	assigned total feeder manhole inv	=IF(basic_segment_type="Cable",subfeeder_dist*BU2*VLOOKUP(line_density,density_inputs,7)*VLOOKUP(line_density,density_inputs,17),(subfeeder_dist+'distribution input'!AO2+'distribution input'!AP2)*BX2*VLOOKUP(line_density,density_inputs,13)*VLOOKUP(line_density,density_inputs,18))+basic_lines*BE2	assigns manhole and pullbox investment to current cluster according to total lines in cluster; includes investment required for subfeeder (with connecting cables, if present)
BG	total feeder u/g placement segment inv per line	=IF(AZ2=0,0,(AR2+AW2)/AZ2)	calculates total underground (conduit) placement investment per line
BH	cumulative u/g placement investment per line	=IF(OR(O1=1,ISTEXT(BH1)),BG2,BG2+BH1)	accumulates feeder (conduit) underground placement investment per line
BI	assigned total feeder u/g placement inv	=IF(basic_segment_type="Cable",subfeeder_dist*BU2*VLOOKUP(line_density,density_inputs,9)*((1-wtg_excav)+wtg_excav*labor_adj),(subfeeder_dist+'distribution input'!AO2+'distribution input'!AP2)*BX2*VLOOKUP(line_density,density_inputs,15)*((1-wtg_excav)+wtg_excav*labor_adj))*(distribution input'!Y2+distribution input'!Z2-1)+basic_lines*BH2	Assigns cluster share of main feeder underground placement investment according to lines in cluster; adds subfeeder placement
BJ	total feeder buried placement segment inv per line	=IF(AZ2=0,0,(AS2+AX2)/AZ2)	calculates buried placement investment for current segment expressed per line
BK	cumulative buried placement investment per line	=IF(OR(O1=1,ISTEXT(BK1)),BJ2,BJ2+BK1)	accumulates buried placement investment per line from beginning of feeder route
BL	assigned total feeder buried placement inv	=IF(basic_segment_type="Cable",subfeeder_dist*BS2*VLOOKUP(line_density,density_inputs,8)*((1-wtg_excav)+wtg_excav*labor_adj),(subfeeder_dist+'distribution input'!AO2+'distribution input'!AP2)*BV2*VLOOKUP(line_density,density_inputs,14)*((1-wtg_excav)+wtg_excav*labor_adj))*(distribution input'!Y2+distribution input'!Z2-1)+basic_lines*BK2	computes cluster's share of total buried placement investment according to lines in cluster; includes subfeeder buried placement and connecting cables for fiber subfeeder, when present
BM	feeder pole segment inv per line	=IF(AZ2=0,0,(AT2+AY2)/AZ2)	calculates feeder pole investment for current segment per line
BN	cumulative inv per line	=IF(OR(O1=1,ISTEXT(BN1)),BM2,BM2+BN1)	Accumulates feeder pole investment per line from beginning of feeder route

Equation Listing

Column	Name	Formula	Description
BO	assigned feeder pole inv	=basic_lines*BN2+IF(basic_segment_type="Cable",1+CEILING(subfeeder_dist*B T2/VLOOKUP(line_density,density_inputs,19,1))*(pole_materials+pole_labor*((1-wtg_pole_set)+wtg_pole_set*labor_adj)*('distribution input'!Y2+'distribution input'!Z2-1)),0)+IF(basic_segment_type<>"Cable",1+CEILING((subfeeder_dist+'distribution input'!AO2+'distribution input'!AP2)*BW2/VLOOKUP(line_density,density_inputs,19,1))*(pole_materials+pole_labor*((1-wtg_pole_set)+wtg_pole_set*labor_adj))*('distribution input'!Y2+'distribution input'!Z2-1),0)	assigns feeder pole investment to cluster according to total lines in cluster; includes subfeeder poles
BP	fiber feeder conduit segment inv per line	=IF(AE2=0,0,AU2/AE2)	Computes per-line investment in conduit containing fiber feeder cable per line
BQ	cumulative inv per line	=IF(OR(O1=1,ISTEXT(BQ1)),BP2,BP2+BQ1)	accumulates fiber conduit investment per line from beginning of feeder route
BR	assigned fiber feeder conduit inv	=IF(basic_segment_type="DLC",basic_lines*BQ2,0)+IF(basic_segment_type="Cab le",0,IF(AO2=0,"",(subfeeder_dist+'distribution input'!AO2+'distribution input'!AP2)*BU2*(conduit_mat_inv_ft*(1+spare_tubes_sect))))	Assigns fiber conduit investment to cluster according to total lines in cluster; includes subfeeder conduit
BS	effective copper buried fraction	=VLOOKUP(line_density,density_inputs,5)+CF2	Computes effective buried structure fraction according to local conditions and user-set input values
BT	effective copper aerial fraction	=MAX(0,1-BS2-BU2)	computes effective aerial structure fraction for copper cable from buried and underground fractions
BU	effective copper u/g fraction	=VLOOKUP(line_density,density_inputs,6)	Repeats user-selected copper cable underground fraction for cluster density range
BV	effective fiber buried fraction	=VLOOKUP(line_density,density_inputs,11)+CN2	Computes effective buried fiber cable structure fraction according to local conditions and user-set input values
BW	effective fiber aerial fraction	=MAX(0,1-BV2-BX2)	computes effective aerial structure fraction for fiber cable from buried and underground fractions
BX	effective fiber u/g fraction	=VLOOKUP(line_density,density_inputs,12)	Repeats user-selected fiber cable underground fraction for cluster density range
BY	copper buried investment/foot, cable + placement, with sharing	=IF(Y2="Cable",((AH2*max_copper_inv+VLOOKUP(AG2,copper_inv,2,FALSE))*inputs!\$C\$57+VLOOKUP(line_density,density_inputs,28))*((1-wtg_copper_const)+wtg_copper_const*labor_adj),0)	Computes total copper buried investment per foot, including placement and structure sharing
BZ	copper aerial investment/foot, cable	=IF(Y2="Cable",((AH2*max_copper_inv+VLOOKUP(AG2,copper_inv,2,FALSE))*((1-wtg_copper_const)+wtg_copper_const*labor_adj)),0)	Calculates aerial copper cable investment per foot

Equation Listing

Column	Name	Formula	Description
CA	copper aerial investment/foot, pole, with sharing	=IF(Y2="Cable", (pole_materials+pole_labor*((1-wtg_pole_set)+wtg_pole_set*labor_adj))/VLOOKUP(line_density,density_inputs,19)*VLOOKUP(line_density,density_inputs,20),0)	Calculates pole investment per foot for copper cables, including effects of pole sharing
CB	std copper buried LC cost/ft, with sharing	=BY2*(LCFactors!\$C\$3+LCFactors!\$D\$3)	Computes life cycle cost/ft of buried cable, including capital carrying costs, maintenance costs, and effects of structure sharing; result represents "standard" cost unaffected by local conditions
CC	std copper aerial LC cost/ft, with sharing	=BZ2*(LCFactors!\$C\$5+LCFactors!\$D\$5)+CA2*(LCFactors!\$C\$9+LCFactors!\$D\$9)	Computes life cycle cost/ft of aerial cable, including capital carrying costs, maintenance costs, and effects of structure sharing; result represents "standard" cost unaffected by local conditions
CD	local copper buried LC cost/ft, w/sharing	=IF(Y2="Cable", ((AH2*max_copper_inv+VLOOKUP(AG2,copper_inv,2,FALSE))*inputs!\$C\$57+('distribution input'!Y2+'distribution input'!Z2-1)*VLOOKUP(line_density,density_inputs,28))*((1-wtg_copper_const)+wtg_copper_const*labor_adj),0)*(LCFactors!\$C\$3+LCFactors!\$D\$3)	Computes life cycle cost/ft of buried cable, including capital carrying costs, maintenance costs, and effects of structure sharing and local rock and difficult surface conditions; result represents local cost adjusted to local conditions
CE	local copper aerial LC cost/ft, w/sharing	=IF(Y2="Cable", BZ2*(LCFactors!\$C\$5+LCFactors!\$D\$5)+(pole_materials+('distribution input'!Y2+'distribution input'!Z2-1)*pole_labor*((1-wtg_pole_set)+wtg_pole_set*labor_adj))/VLOOKUP(line_density,density_inputs,19)*VLOOKUP(line_density,density_inputs,20)*(LCFactors!\$C\$9+LCFactors!\$D\$9),0)	Computes life cycle cost/ft of aerial cable, including capital carrying costs, maintenance costs, and effects of structure sharing and local rock and difficult surface conditions; result represents "local" cost adjusted for local conditions
CF	copper buried adjustment	=IF(Y2="Cable", (0.5-1/(1+(CB2/CC2)/((CD2/CE2)^inputs!\$G\$82)))*VLOOKUP(line_density,density_inputs,23),0)	Calculates adjustment to input buried structure fraction for local density range using user-specified buried "swing" factor and standard and local cost inputs to logistic function
CG	fiber buried investment/foot, cable + placement, with sharing	=IF(Y2="DLC", VLOOKUP(AL2,fiber_inv,2,FALSE)+inputs!\$C\$58+AM2*(max_fiber_inv+inputs!\$C\$58)+VLOOKUP(line_density,density_inputs,29),0)	Computes total fiber buried investment per foot, including placement and structure sharing
CH	fiber aerial investment/foot, cable	=IF(Y2="DLC", VLOOKUP(AL2,fiber_inv,2,FALSE)+AM2*(max_fiber_inv),0)	Calculates aerial fiber cable investment per foot
CI	fiber aerial investment/foot, pole, with sharing	=IF(Y2="DLC", (pole_materials+pole_labor*((1-wtg_pole_set)+wtg_pole_set*labor_adj))/VLOOKUP(line_density,density_inputs,19)*VLOOKUP(line_density,density_inputs,24),0)	Calculates pole investment per foot for fiber cables, including effects of pole sharing
CJ	std fiber buried LC cost/ft, with sharing	=CG2*(LCFactors!\$C\$4+LCFactors!\$D\$4)	Computes life cycle cost/ft of buried cable, including capital carrying costs, maintenance costs, and effects of structure sharing; result represents "standard" cost unaffected by local conditions

Equation Listing

Column	Name	Formula	Description
CK	std fiber aerial LC cost/ft, with sharing	=CH2*(LCFactors!\$C\$6+LCFactors!\$D\$6)+CI2*(LCFactors!\$C\$9+LCFactors!\$D\$9)	Computes life cycle cost/ft of aerial cable, including capital carrying costs, maintenance costs, and effects of structure sharing; result represents "standard" cost unaffected by local conditions
CL	local fiber buried LC cost/ft, w/sharing	=IF(Y2="DLC",VLOOKUP(AL2,fiber_inv,2,FALSE)+inputs!\$C\$58+AM2*(max_fiber_inv+inputs!\$C\$58)+('distribution input'!Y2+'distribution input'!Z2-1)*VLOOKUP(line_density,density_inputs,29),0)*(LCFactors!\$C\$4+LCFactors!\$D\$4)	Computes life cycle cost/ft of buried cable, including capital carrying costs, maintenance costs, and effects of structure sharing and local rock and difficult surface conditions; result represents local cost adjusted to local conditions
CM	local fiber aerial LC cost/ft, w/sharing	=IF(Y2="DLC",CH2*(LCFactors!\$C\$6+LCFactors!\$D\$6)+(pole_materials+('distribution input'!Y2+'distribution input'!Z2-1)*pole_labor*((1-wtg_pole_set)+wtg_pole_set*labor_adj))/VLOOKUP(line_density,density_inputs,19)*VLOOKUP(line_density,density_inputs,24)*(LCFactors!\$C\$9+LCFactors!\$D\$9),0)	Computes life cycle cost/ft of aerial cable, including capital carrying costs, maintenance costs, and effects of structure sharing and local rock and difficult surface conditions; result represents "local" cost adjusted for local conditions
CN	fiber buried adjustment	=IF(Y2="DLC", (0.5-1/(1+(CJ2/CK2)/((CL2/CM2)^inputs!\$G\$82)))*VLOOKUP(line_density,density_inputs,27),0)	Calculates adjustment to input fiber buried structure fraction for local density range using user-specified buried "swing" factor and standard and local cost inputs to logistic function

Equation Listing

Column	Name	Formula	Description
A	wire center	=distribution input!C2	repeats wire center location ID from distribution input sheet
B	operating company	=distribution input!A2	repeats company name from distribution input sheet
C	CBG	=distribution input!D2	repeats CBG from distribution input sheet
D	operating company indicator	=distribution input!B2	repeats operating company type indicator from distribution input sheet
E	total lines	=distribution input!I2	repeats total lines from distribution input sheet
F	business lines	=distribution input!L2	repeats business lines from distribution input sheet
G	res lines	=distribution input!M2	repeats residential line total from distribution input sheet
H	SA lines	=distribution input!N2	repeats special access line total from distribution input sheet
I	public lines	=distribution input!O2	repeats public line total from distribution input sheet
J	households	=distribution input!P2	repeats household total from distribution input sheet
K	single-line business lines	=distribution input!Q2	repeats single-line business line total from distribution input sheet
L	area, sq mi	=distribution input!K2	repeats area from distribution input sheet
M	density lines/sq mi	=distribution input!J2	repeats line density from distribution input sheet
N	density range	=VLOOKUP(M2,density_inputs,16)	repeats density range from distribution input sheet
O	cpr fdr cbl inv, u/g	=('cable inv'\$R2+'cable inv'!J2)*'cable inv'!BU2+IF('cable inv'!H2="N/A",0,'cable inv'!D2*VLOOKUP('cable inv'!H2,copper_inv,2,FALSE)*'cable inv'!BU2)	computes overall feeder copper cable investment for underground placement, including effect of regional labor multiplier
P	cpr fdr cbl inv, buried	=('cable inv'\$R2+'cable inv'!J2)*'cable inv'!BS2*inputs!\$C\$57+IF('cable inv'!H2="N/A",0,'cable inv'!D2*VLOOKUP('cable inv'!H2,copper_inv,2,FALSE)*'cable inv'!BS2)*inputs!\$C\$57	computes overall feeder copper cable investment for buried placement, including effect of regional labor multiplier
Q	cpr fdr cbl inv, aerial	=('cable inv'\$R2+'cable inv'!J2)*'cable inv'!BT2+IF('cable inv'!H2="N/A",0,'cable inv'!D2*VLOOKUP('cable inv'!H2,copper_inv,2,FALSE)*'cable inv'!BT2)	computes overall feeder copper cable investment for aerial placement, including effect of regional labor multiplier
R	fiber fdr cbl inv, u/g	=('cable inv'\$U2+'cable inv'!N2)*'cable inv'!BX2	computes overall feeder fiber cable investment for underground placement
S	fiber fdr cbl inv, buried	=('cable inv'\$U2+'cable inv'!N2+IF('cable inv'!Y2<>"Cable",('cable inv'!D2+'cable inv'!X2+'distribution input'!AM2+'distribution input'!AN2)*inputs!\$C\$58,0))*'cable inv'!BV2	computes overall feeder fiber cable investment for buried placement,

Equation Listing

Column	Name	Formula	Description
T	fiber fdr cbl inv aerial	=("cable inv"!\$U2+"cable inv"!N2)*"cable inv"!BW2	computes overall feeder fiber cable investment for aerial placement
U	fdr conduit inv	="cable inv"!BC2+"cable inv"!BR2	calculates total conduit investment as sum of copper and fiber investments
V	feeder manhole inv	="cable inv"!BF2	repeats feeder manhole investment (includes pullboxes for fiber)
W	feeder u/g copper plcmnt inv	=IF("cable inv"!\$Y2="Cable",'cable inv'!\$BI2,0)	repeats underground copper placement investment
X	feeder u/g fiber plcmnt inv	=IF("cable inv"!\$Y2="DLC",'cable inv'!\$BI2,0)	repeats feeder underground fiber placement investment
Y	feeder buried copper plcmnt inv	=IF("cable inv"!\$Y2="Cable",'cable inv'!\$BL2,0)	repeats feeder buried copper placement investment
Z	feeder buried fiber plcmnt inv	=IF("cable inv"!\$Y2="DLC",'cable inv'!\$BL2,0)	repeats feeder buried fiber placement investment
AA	feeder pole inv	="cable inv"!BO2	repeats feeder pole investment
AB	dist cable inv, underground	="distribution input"!R2	repeats distribution underground cable investment
AC	dist cable inv, buried	="distribution input"!S2	repeats buried distribution cable investment
AD	dist cable inv, aerial	="distribution input"!T2	repeats distribution aerial cable investment
AE	distribution conduit inv	="distribution input"!U2	repeats distribution conduit investment
AF	distribution conduit plcmnt inv	="distribution input"!V2	repeats distribution underground conduit placement investment
AG	dist buried plcmnt inv	="distribution input"!X2	repeats distribution buried placement investment
AH	dist pole inv	="distribution input"!W2	repeats distribution pole investment
AI	calc cpr fdr fill	="cable inv"!AF2	repeats calculated copper feeder achieved fill at cluster
AJ	calc dist fill	="distribution input"!AB2	repeats calculated distribution achieved fill
AK	calc "mainframe" fill	=IF(AND('cable inv'!W2=1,'cable inv'!Y2="Cable"),output!AI2,0)	selects achieved copper feeder fill at first cluster along copper feeder cable route; equivalent to fill at mainframe
AL	DLC inv w/site	="distribution input"!AD2+"distribution input"!AF2	repeats DLC investment
AM	SAI inv	="distribution input"!AH2	repeats SAI investment
AN	terminal inv	="distribution input"!AI2	repeats terminal investment
AO	drop inv	="distribution input"!AJ2	repeats drop investment
AP	NID inv	="distribution input"!AK2	repeats NID investment

Workbook: R50A_feeder.xls
Worksheet: output

Equation Listing

HAI Model, v5.0A
Feeder Module

Column	Name	Formula	Description
AQ	feeder distance	='cable inv'!X2+'cable inv'!D2	calculates incremental distance from previous cluster plus subfeeder distance
AR	total dist distance	='distribution input'!H2	repeats total distribution structure distance
AS	DLC lines	=IF('cable inv'!Y2="DLC",'cable inv'!E2,0)	repeats number of lines served by DLC
AT	wtd cluster average loop length	='distribution input'!BA2*'distribution input'!BB2	Calculates average loop length for cluster weighted by fraction of total wire center lines in cluster for use by interface in computing weighted average loop length for wire center
AU	cluster serial number	='distribution input'!AT2	repeats cluster serial number from distribution input sheet

Equation Listing

HAI Model, v5.0A
 Switching/Interoffice Module

Column	Name	Formula	Description
A	Host		First three columns are populated by interface when host/remote calculations are enabled and remote/host assignments made; these entries function as a lookup table for wire center investment sheet to obtain host and remote investments
B	Remote		
C	NECA Co Code		
D	total local switched lines per host	=IF(\$A1<>\$A2,IF(ISNA(VLOOKUP(\$A2,inv_tbl,1,FALSE)),0,E2+F2),0)	calculates switched line total served from host wire center
E	total local residential lines per host	=IF(\$A1<>\$A2,IF(ISNA(VLOOKUP(\$A2,inv_tbl,1,FALSE)),0,VLOOKUP(\$A2,inv_tbl,4,FALSE)/VLOOKUP(\$A2,inv_tbl,2,FALSE)),0)	calculates total residential lines served from host wire center
F	total local business + public lines per host	=IF(\$A1<>\$A2,IF(ISNA(VLOOKUP(\$A2,inv_tbl,1,FALSE)),0,VLOOKUP(\$A2,inv_tbl,5,FALSE)/VLOOKUP(\$A2,inv_tbl,2,FALSE)),0)	calculates total business and public lines served from host wire center
G	total HR ring traffic per host, CCS	=IF(A1<>A2,(host_res*res_loc_dir+host_bus*bus_loc_dir)*hr_fraction,0)	calculates total traffic contributed to local host/remote ring by host
H	total BHCA per host	=IF(\$A1<>\$A2,IF(ISNA(VLOOKUP(\$A2,inv_tbl,1,FALSE)),0,VLOOKUP(\$A2,inv_tbl,65,FALSE)/VLOOKUP(\$A2,inv_tbl,2,FALSE)),0)	calculates total interoffice BHCA generated by host alone
I	total interoffice traffic per host, CCS	=IF(\$A1<>\$A2,IF(ISNA(VLOOKUP(\$A2,inv_tbl,1,FALSE)),0,VLOOKUP(\$A2,inv_tbl,66,FALSE)/VLOOKUP(\$A2,inv_tbl,2,FALSE)),0)*io_fraction	calculates total interoffice busy hour traffic generated by host alone
J	total switched lines per remote	=IF(ISNA(VLOOKUP(\$B2,inv_tbl,1,FALSE)),0,K2+L2)	calculates switched line total served by remote
K	total residential lines per remote	=IF(ISNA(VLOOKUP(\$B2,inv_tbl,1,FALSE)),0,VLOOKUP(\$B2,inv_tbl,4,FALSE)/VLOOKUP(\$B2,inv_tbl,2,FALSE))	calculates total residential lines served by remote
L	total business + public lines per remote	=IF(ISNA(VLOOKUP(\$B2,inv_tbl,1,FALSE)),0,VLOOKUP(\$B2,inv_tbl,5,FALSE)/VLOOKUP(\$B2,inv_tbl,2,FALSE))	calculates total business and public lines served by remote
M	cumulative BHCA	=IF(A3=A2,M3+H2+N2,H2+N2)	accumulates busy-hour call attempts in host/remote system
N	total BHCA per remote	=IF(ISNA(VLOOKUP(\$B2,inv_tbl,1,FALSE)),0,VLOOKUP(\$B2,inv_tbl,65,FALSE)/VLOOKUP(\$B2,inv_tbl,2,FALSE))	total BHCA generated by remote
O	total interoffice traffic per remote, CCS	=IF(ISNA(VLOOKUP(\$B2,inv_tbl,1,FALSE)),0,VLOOKUP(\$B2,inv_tbl,66,FALSE)/VLOOKUP(\$B2,inv_tbl,2,FALSE))*io_fraction	total interoffice busy hour traffic generated by remote, including remote-host traffic
P	switch inv per host	=IF(\$A2<>\$A1,VLOOKUP(\$W2,sw_inv_tbl,IF(OR('loop db inputs'!\$B\$2=8,'loop db inputs'!\$B\$2=1),3,9))+D2/line_fill*VLOOKUP(\$W2,sw_inv_tbl,IF(OR('loop db inputs'!\$B\$2=8,'loop db inputs'!\$B\$2=1),6,12))-W2/6*inputs!\$C\$37+AL2*inputs!\$C\$37+AQ2*inputs!\$C\$97/2-Z2*inputs!\$C\$24,0)*sw_install_mult	total sw inv per host, using total lines in system for table entry and directly-served lines for variable investment calculation

Equation Listing

Column	Name	Formula	Description
Q	total per line wire center inv per system	=IF(A1<>A2,AY2/W2,0)	calculates total average wire center investment per line for host/remote system
R	switch inv per remote	=(VLOOKUP(\$J2,sw_inv_tbl,IF(OR('loop db inputs'!\$B\$2=8,'loop db inputs'!\$B\$2=1),4,10))+J2/line_fill*VLOOKUP(\$J2,sw_inv_tbl,IF(OR('loop db inputs'!\$B\$2=8,'loop db inputs'!\$B\$2=1),7,13))-AA2*inputs!\$C\$24)*sw_install_mult	computes remote switch investment
S	cumulative switch inv per system	=IF(A3=A2,S3+P2+R2,P2+R2)	accumulates total switching investment for host/remote system
T	repeated wire center inv per line	=IF(A1<>A2,Q2,T1)	repeats wire center investment per line in all records for host/remote system
U	avg switch inv per line in system	=IF(A1<>A2,S2/W2,0)	computes overall average switch investment per line for host/remote system
V	repeated average switch inv per line	=IF(A2<>A1,U2,V1)	repeats average switch investment per line for all records in host/remote system
W	total lines in system	=X2+Y2	calculates total lines in host/remote system
X	total residential lines in system	=IF(A3=A2,X3+E2+K2,E2+K2)	calculates total residential lines in host/remote system
Y	total business + public lines in system	=IF(A3=A2,Y3+F2+L2,F2+L2)	calculates total business + public lines in host/remote system
Z	DLC lines per host wire center	=IF(\$A1<>\$A2,VLOOKUP(A2,inv_tbl,57,FALSE),0)	computes total DLC lines served from host wire center
AA	DLC lines per remote wire center	=VLOOKUP(B2,inv_tbl,57,FALSE)	calculates total DLC lines served by remote
AB	repeated HR ring term inv/line	=IF(A2<>A1,BH2,AB1)	repeats terminal investment per line for host/remote ring for all records in system
AC	cumulative local direct traffic, CCS	=IF(\$A3=\$A2,AC3+((host_res*(1-hr_fraction)+remote_res*(1-rh_fraction))*res_loc_dir+(host_bus*(1-hr_fraction)+remote_bus*(1-rh_fraction))*bus_loc_dir)*0.5,((host_res*(1-hr_fraction)+remote_res*(1-rh_fraction))*res_loc_dir+(host_bus*(1-hr_fraction)+remote_bus*(1-rh_fraction))*bus_loc_dir)*0.5)	accumulates local direct-routed traffic for host/remote system
AD	total local direct trunks per host	=IF(\$A1<>\$A2,IF(AC2<=trfc_thresh,VLOOKUP(AC2,trk_table,2),CEILING(AC2/trk_occ,1)),0)	computes local direct trunks required per host from traffic table
AE	cumulative local tandem traffic, CCS	=IF(\$A3=\$A2,AE3+(host_res+remote_res)*res_loc_tdm+(host_bus+remote_bus)*bus_loc_tdm,(host_res+remote_res)*res_loc_tdm+(host_bus+remote_bus)*bus_loc_tdm)	accumulates local tandem-routed traffic for host/remote system

Equation Listing

Column	Name	Formula	Description
AF	total local tandem trunks per host	=IF(\$A1>\$A2,IF(AE2<=trfc_thresh,VLOOKUP(AE2,trk_tablec,2),CEILING(AE2/trk_occ,1)),0)	computes local tandem trunks required per host from traffic table
AG	cumulative intraLATA direct traffic, CCS	=IF(\$A3=\$A2,AG3+((host_res+remote_res)*res_LATA_dir+(host_bus+remote_bus)*bus_LATA_dir)*0.5,((host_res+remote_res)*res_LATA_dir+(host_bus+remote_bus)*bus_LATA_dir)*0.5)	accumulates intraLATA direct-routed traffic for host/remote system
AH	total intraLATA direct trunks per host	=IF(\$A1>\$A2,IF(AG2<=trfc_thresh,VLOOKUP(AG2,trk_table,2),CEILING(AG2/trk_occ,1)),0)	computes intraLATA direct trunks trunks required per host from traffic table
AI	cumulative intraLATA tandem traffic, CCS	=IF(\$A3=\$A2,AI3+(host_res+remote_res)*res_LATA_tdm+(host_bus+remote_bus)*bus_LATA_tdm,(host_res+remote_res)*res_LATA_tdm+(host_bus+remote_bus)*bus_LATA_tdm)	accumulates intraLATA tandem-routed traffic for host/remote system
AJ	total intraLATA tandem trunks per host	=IF(\$A1>\$A2,IF(AI2<=trfc_thresh,VLOOKUP(AI2,trk_table,2),CEILING(AI2/trk_occ,1)),0)	computes intraLATA tandem trunks trunks required per host from traffic table
AK	cumulative OS traffic, CCS	=IF(\$A3=\$A2,AK3+(host_res+remote_res)*res_OS+(host_bus+remote_bus)*bus_OS,(host_res+remote_res)*res_OS+(host_bus+remote_bus)*bus_OS)	accumulates operator services traffic for host/remote system
AL	total OS trunks per host	=IF(\$A1>\$A2,IF(AK2<=trfc_thresh,VLOOKUP(AK2,trk_table,2),CEILING(AK2/trk_occ,1)),0)	computes operator services trunks trunks required per host from traffic table
AM	cumulative direct-routed access traffic, CCS	=IF(\$A3=\$A2,AM3+(host_res+remote_res)*res_acc_dir+(host_bus+remote_bus)*bus_acc_dir,(host_res+remote_res)*res_acc_dir+(host_bus+remote_bus)*bus_acc_dir)	accumulates direct-routed access traffic for host/remote system
AN	total direct-routed access trunks per host	=IF(\$A1>\$A2,IF(AM2<=trfc_thresh,VLOOKUP(AM2,trk_table,2),CEILING(AM2/trk_occ,1)),0)	computes direct access trunks trunks required per host from traffic table
AO	cumulative tandem-routed access traffic, CCS	=IF(\$A3=\$A2,AO3+(host_res+remote_res)*res_acc_tdm+(host_bus+remote_bus)*bus_acc_tdm,(host_res+remote_res)*res_acc_tdm+(host_bus+remote_bus)*bus_acc_tdm)	accumulates tandem-routed access traffic for host/remote system
AP	total tandem-routed access trunks per host	=IF(\$A1>\$A2,IF(AO2<=trfc_thresh,VLOOKUP(AO2,trk_table,2),CEILING(AO2/trk_occ,1)),0)	computes tandem access trunks trunks required per host from traffic table
AQ	total A links per host	=IF(A1>A2,2*CEILING(M2*inputs!\$F\$63,1),0)	computes total A signaling links per host
AR	SA lines per host	=IF(A1>A2,VLOOKUP(A2,loop_in_tbl,8),0)	calculates total special access lines per host
AS	SA lines per remote	=VLOOKUP(B2,loop_in_tbl,8)	calculates total special access lines per remote
AT	cumulative SA lines	=IF(A3=A2,AT3+AS2,AS2)	accumulates special access lines in host/remote system
AU	total SA lines per system	=IF(A1>A2,AR2+AT2,0)	calculates total special access lines for host/remote system
AV	total switched trunks per host	=AP2+AN2+AL2+AJ2+AH2+AF2+AD2	calculates total switched trunks per host
AW	wire center inv per host	=IF(\$A1>\$A2,VLOOKUP(\$D2,wc_inv,7),0)	computes wire center investment for host wire center

Equation Listing

Column	Name	Formula	Description
AX	wire center inv per remote	=VLOOKUP(\$J2,we_inv,7)	computes wire center investment for remote wire center
AY	cumulative wire center inv	=IF(A3=A2,AY3+AW2+AX2,AW2+AX2)	accumulates wire center investment for host/remote system
AZ	host ADM inv -- HR ring	=IF(A1>A2,IF(BF2=0,inputs!\$C\$159+(CEILING((BC2+BD2)*(1+transit_fac)/2/inputs!\$C\$165/7,1)-12)*inputs!\$C\$149,IF(CEILING((BC2+BD2)*(1+transit_fac)/2/inputs!\$C\$165/28,1)<=12,inputs!\$C\$158,CEILING((BC2+BD2)*(1+transit_fac)/2/inputs!\$C\$165/28/max_rate,1)*inputs!\$C\$157))+CEILING((BC2+BD2)*(1+transit_fac)/2/inputs!\$C\$165/28,1)*inputs!\$C\$164,0)	computes transmission terminal investment at host location for host/remote ring
BA	remote ADM inv -- HR ring	=IF(BF2=0,inputs!\$C\$159+(CEILING((BB2)*(1+transit_fac)/2/inputs!\$C\$165/7,1)-12)*inputs!\$C\$149,IF(CEILING(BB2*(1+transit_fac)/2/inputs!\$C\$165/28,1)<=12,inputs!\$C\$158,CEILING(BB2*(1+transit_fac)/2/inputs!\$C\$165/28/max_rate,1)*inputs!\$C\$157))+CEILING(BB2*(1+transit_fac)/2/inputs!\$C\$165/28,1)*inputs!\$C\$164)	computes transmission terminal investment for remote wire center
BB	total ring DS0s per remote	=IF(O2<trfc_thresh,VLOOKUP(O2,trk_table,2),CEILING(O2/trk_occ,1))	calculates total DS0s on ring for remote
BC	total HR ring DS0s, host trfc only	=IF(A1>A2,IF(G2<trfc_thresh,VLOOKUP(G2,trk_table,2),CEILING(G2/trk_occ,1)),0)	calculates total DS-0s contributed to host/remote ring by host
BD	cumulative remote DS0s	=IF(A3=A2,BD3+BB2,BB2)	accumulates ring DS0s for host/remote system
BE	> OC3 determination	=IF(A1>A2,IF(CEILING((BD2+BC2)*(1+transit_fac)/2/inputs!\$C\$165/28,1)>3,1,0),0)	indicates whether ring capacity exceeds OC-3
BF	HR ring > OC3 ind	=IF(A1>A2,BE2,BF1)	repeats ring capacity indication
BG	cumulative HR ring terminal investment	=IF(A3=A2,BG3+AZ2+BA2,AZ2+BA2)	accumulates investment in terminal equipment for host/remote ring
BH	HR ring terminal inv per line	=IF(A1>A2,BG2/W2,0)	computes average investment per line in all host/remote ring terminal equipment

Workbook: R50A_switching_io.xls
Worksheet: ring io

Equation Listing

HAI Model, v5.0A
Switching/Interoffice Module

Column	Name	Formula	Description
A	Remote		
B	Remote Vert (NECA)		
C	Remote Horiz (NECA)		
D	NECA Co Code		
E	Host		
F	Host Vert (NECA)		
G	Host Horiz (NECA)		
H			
I	Wire Center		
J	WC Vert (NECA)		
K	WC Horiz (NECA)		
L	NECA Co Code		
M	Tandem		
N	Tandem Vert (NECA)		
O	Tandem Horiz (NECA)		
P			
Q	Remote		
R	Remote Connects to CLLI (CLLI #1)		
S	Distance From Remote to CLLI #1, mi.		
T	Remote Connects to CLLI (CLLI #2)		
U	Distance from Remote to CLLI #2, mi.		
V	Ring Connector Node #1		
W	Ring Connector Node #2		
X	Ring Connector Distance, mi		
Y			
Z	Wire Center		
AA	Wire Center Connects to CLLI (CLLI #1)		
AB	Distance from Wire Center to CLLI #1, mi.		

Equation Listing

Column	Name	Formula	Description
AC	Wire Center Connects to CLLI (CLLI #2)		
AD	Distance from Wire Center to CLLI #2, mi.		
AE	DS-3 Equivalents		
AF	DS-3 Equivalents from Spur(s)		
AG	Ring Connector Node #1		
AH	Ring Connector Node #2		
AI	Ring Connector Distance, mi.		
AJ	Total Ring Connector Distance (mi)		
AK	Total Number of Ring Connectors		
AL			
AM	CLLI		
AN	Distance (mi)		
AO	DS-3 Equivalents		
AP	DS-3 Equivalents from Spur(s)		
AQ			
AR	Spur-Connected CLLI		
AS	Spur Connects To CLLI		
AT	Spur Distance, mi.		
AU	Spur CLLI DS-3 Equivalents		
AV			
AW	Ring System Interconnection CLLI #1		
AX	CLLI #1 Homes on Tandem		
AY	CLLI #1 Connects to CLLI (CLLI #2)		
AZ	CLLI # 2 Homes on Tandem		
BA	Ring System Interconnector Distance, mi.		
BB			

Equation Listing

Column	Name	Formula	Description
A	wire center		
B	STP A link distance sum		
C	local tandem distance		
D	OS Tandem Distance		
E	Ring Distance		
F	NECA Company Code		
G	NECA Vert Coord		
H	NECA Horiz Coord		
I	Serving Tandem		
J	Tandem NECA Company Code		
K	NECA Tandem Vert Coord		
L	NECA Tandem Horiz Coord		
M	Tandem LATA (From NECA Data)		
N	WC Connects to BOC CLLI		
O	Total DS-3 Equivalents in Ring		
P			
Q	Company Code		
R	Total tandems in study area		
S	Total OS tdms in study area		
T	Total tandem/STP A-Link distance		
U	Total STP pairs in study area		
V	Total STP/STP distance		
W	Total Tandem Mesh Distance		
X	Total Inter-Ring Distance		
Y	Total Number of Ring Connectors		
Z	Total Inter-Ring System Distance		
AA	Total Number of Inter-Ring System Connectors		

Workbook: R50A_switching_io.xls
Worksheet: distance inputs

Equation Listing

HAI Model, v5.0A
Switching/Interoffice Module

Column	Name	Formula	Description
AB	Total Number of Rings Intersecting a Tandem		
AC	total unidentified tdm distance		

Equation Listing

Column	Name	Formula	Description
A	wire center		
B	operating company indicator		
C	area, sq mi		
D	total lines		
E	business lines		
F	res lines		
G	public lines		
H	SA lines		
I	DLC lines		
J	feeder pole inv		
K	feeder buried fiber plcmt inv		
L	feeder buried copper plcmt inv		
M	feeder u/g fiber plcmt inv		
N	feeder u/g copper plcmt inv		
O	feeder manhole inv		

Equation Listing

Row	B	C	D	Description
50				
51		total links	=SUMPRODUCT('wire center investment'!F2:F2200,'wire center investment'!Z2:Z2200)	
52		total link investment	=SUMPRODUCT('wire center investment'!F2:F2200,'wire center investment'!AA2:AA2200)	
53		average link inv	=IF(D51=0,0,D52/D51)	
54		total tandem A links	=2*(D4+H7)	
55		total C links	=4*(D18)*inputs!C101	
56		equiv tdm A links/C links/linc	=(D54+D55)/(D5+D6+D7)	
57				
58				
59	Total SCP investment per line		=inputs!F61	
60	Total SCP wire center investment per line		=(inputs!\$E\$139*inputs!\$D\$139+inputs!\$C\$139)/(D5+D6+D7)	
61				
62	Average ring distance per node, mi		=IF(COUNT('ring io'!AN:AN)=0,0,SUM('ring io'!AN:AN)/COUNT('ring io'!AN:AN))	
63	Average tandem distance, mi		=SUM('distance inputs'!C:C)/COUNT('distance inputs'!C:C)	
64	Ring + interconnector distance adjustment factor		=IF(SUM('ring io'!AN:AN)-'distance inputs'!AC2=0,0,(SUM('ring io'!AN:AN)-'distance inputs'!X\$2+'distance inputs'!\$Z\$2)/(SUM('ring io'!AN:AN)-'distance inputs'!AC2))	This calculation produces an adjustment factor applied to ring distances to accommodate the additional distance covered by inter-ring connections; it is applied to the ring distance calculation in the wire center investment sheet

Equation Listing

Row	F	G	H	Description
7	number of operator tandems		=VLOOKUP(\$D\$44,tdm_tbl,3,FALSE)	
8	total operator traffic, CCS		=D5*inputs!F91+'tandem and STP investment'!D6*inputs!F80	
9	total operator DS-3s		=H8/trk_occ/28	
10				
11	total operator positions		=H8/(inputs!C114*inputs!C115)	
12				
13	total OS tdm common equipment		=H7*inputs!C86	
14				
15	total OS tdm, per line		=H13/(D5+D6+D7)	
16				
17	total operator position investment		=H11*inputs!C113	
18				
19	total operator pos. investment/line		=H17/(D6+D5+D7)	
20				
21				
22	total OS tdm wire center		=H7*(inputs!\$E\$141*inputs!\$D\$141+inputs!\$C\$141)	
23				
24	total OS tdm wire center, per line		=H22/(D5+D6+D7)	
25				
26				
27				
28				
29	total additional bridge ADMs required		=4*'distance inputs'!Y2+2*'distance inputs'!AA2	The calculations in H29 - H35 compute investment in ADMs and DCSs for tandems and OS tandems as well as for inter-ring connections to produce an overall common ADM/DCS investment (H30) per line added to all lines in study area

Equation Listing

Row	F	G	H	Description
30	total added ADM and DCS investment per line		=IF(H29<=0,0,(H29*inputs!C157+H31+H32+H34+H35)/(tandem and STP investment!D5+'tandem and STP investment'!D6+'tandem and STP investment'!D7+D9))	
31	total tandem ADM inv per tdm loc		=IF(\$D\$4=0,0,inputs!\$C\$157*distance inputs!AB2)	
32	total tandem DCS inv per tdm loc		=IF(\$D\$4=0,0,\$D\$4*inputs!\$C\$164*CEILING(\$D\$10/\$D\$4,1))	
33	average interoffice distance, mi		=SUMPRODUCT(output!C2:C5000,'wire center investment'!B02:B5000)/SUM(output!C2:C5000)	
34	total OS tdm ADM inv per loc		=IF(OS_tdm_count=0,0,OS_tdm_count*inputs!\$C\$157*CEILING(\$H\$9/48/OS_tdm_count,1))	
35	total OS tdm DCS inv per loc		=IF(OS_tdm_count=0,0,OS_tdm_count*inputs!\$C\$164*CEILING(\$H\$9/OS_tdm_count,1))	
36				
37	entrance facility calculations			The calculations in this section develop investment in entrance facilities, including terminal equipment, cable, and structure; they apply only to BOCs and large ICOs (operating company types 8 and 1)
38				
39		terminal multiplexer, per line	=IF(H50=0,0,H50*CEILING((H47+H48)/672/inputs!C165/H50/48,1)*inputs!C157/(D5+D6+D7+D9))	
40		cable investment, per line	=inputs!C38*H49*inputs!C192/(D5+D6+D7+D9)	
41		u/g placement, per line	=\$H\$50*inputs!\$C\$38*inputs!\$E\$196/(\$D\$5+\$D\$6+\$D\$7+\$D\$9)	
42		buried placement, per line	=\$H\$50*inputs!\$C\$38*inputs!\$E\$195/(\$D\$5+\$D\$6+\$D\$7+\$D\$9)	
43		pole inv, per line	=\$H\$50*inputs!\$C\$38*inputs!\$E\$194/(\$D\$5+\$D\$6+\$D\$7+\$D\$9)	
44		pullbox inv, per line	=\$H\$50*inputs!\$C\$38*inputs!\$E\$197/(\$D\$5+\$D\$6+\$D\$7+\$D\$9)	
45		conduit inv, per line	=\$H\$50*inputs!\$C\$38*inputs!\$E\$198/(\$D\$5+\$D\$6+\$D\$7+\$D\$9)	
46		total per line e.f. investment	=IF(tdm_count>0,SUM(H39:H45),0)	
47		total SA lines	=D9	

Equation Listing

Row	F	G	H	Description
48		total switched access trunks	=SUM('wire center investment'!\$BS\$2:\$BS\$5000)- 'tandem and STP investment'!H47	
49		total OC-48s, w/fill	=CEILING((H47+H48)/inputs!C165/672/48,1)	
50		no. of entrance facilities	=D4*inputs!C40	

Equation Listing

Column	Name	Formula	Description
A	location	=`loop db inputs`!A2	repeats wire center location ID
B	switches required	=IF(F2=0,0,MAX(0,CEILING(F2/line_fill/inputs!\$C\$17,1),CEILING((BM2*IF(`loop db inputs`!E2+`loop db inputs`!G2)/F2<inputs!\$C\$22,inputs!\$C\$20/inputs!\$C\$19,inputs!\$C\$20+(inputs!\$C\$21-inputs!\$C\$20)*(`loop db inputs`!E2+`loop db inputs`!G2)/F2)-inputs!\$C\$22)/(1-inputs!\$C\$22))/inputs!\$C\$19)/VLOOKUP(F2,sw_capacity,2),1),CEILING(BN2/VLOOKUP(F2,sw_capacity,3),1)))	computes number of switches required in wire center by considering switch port, real time, and traffic limits
C	total lines	=`loop db inputs`!D2	Repeats total lines, including switched and special access, served by wire center
D	total residential lines	=`loop db inputs`!F2	repeats total residential lines from loop db input sheet
E	total business + public lines	=`loop db inputs`!E2+`loop db inputs`!G2	calculates sum of business and public lines in wire center
F	total switched lines	=`loop db inputs`!E2+`loop db inputs`!F2+`loop db inputs`!G2	calculates total switched lines (residential + business + public) in wire center
G	host/remote indicator (user defined)	=IF(AND(COUNTA('host remote'!A:A)>1,hr_enable),IF(ISNA(VLOOKUP(A2,host_list,1,FALSE)),IF(ISNA(VLOOKUP(A2,remote_tbl,1,FALSE)),"A","R"),"H"),""))	Indicates switch type according to user-invoked options: H = host R = remote A = autonomous blank = aggregated investment selected
H	installed EO switching per line	=IF(AND(sw_type="H",B2>1),(I-1/B2)*BU2+BV2/B2,BU2+BV2+BW2+BX2)	calculates end office switching investment per line according to switch type
I	MDF/protector investment per line	=IF(`loop db inputs`!D2=0,0,inputs!\$C\$23*(`loop db inputs`!D2-`loop db inputs`!I2)/`loop db inputs`!D2)	calculates total main distribution frame and protector investment per line, with adjustment for DLC-served lines (which do not require MDF/protector investment in wire center)
J	end office wire center per line	=IF(F2=0,0,IF(OR(sw_type="",sw_type="A"),1/F2*(VLOOKUP(`wire center investment`!F2,wc_inv,7)+IF(B2>1,B2*VLOOKUP(F2/B2,wc_inv,6),0)),IF(sw_type="R",BZ2,IF(AND(sw_type="H",B2>1),BZ2+B2/F2*VLOOKUP(F2/B2,wc_inv,6),BZ2))))	calculates per-line investment in wire center facilities
K	total local direct-routed traffic, CCS	=(bus_public_lines*bus_loc_dir+res_lines*res_loc_dir)*0.5	computes total offered load for wire center for local direct-routed traffic
L	total local direct trunks required (equiv per line)	=IF(\$C2=0,0,1/\$C2*IF(G2="H",VLOOKUP(A2,hr_tbl,30,FALSE),IF(K2<=trfc_thresh,VLOOKUP(K2,trk_table,2),CEILING(K2/inputs!\$C\$36,1))))	computes total local direct trunks required according to total offered load calculation and user-set inputs for maximum trunk occupancy

Equation Listing

Column	Name	Formula	Description
M	local direct trunk investment per line	=IF(\$C2=0,0,\$C2*L2/\$AR2*\$AU2)	calculates share of total interoffice facility investment assigned to local direct trunks
N	total local tandem-routed traffic, CCS	=bus_public_lines*bus_loc_tdm+res_lines*res_loc_tdm	computes total offered load for wire center for local tandem-routed traffic
O	total local tdm trks required (equiv per line)	=IF(\$C2=0,0,1/\$C2*IF(G2="H",VLOOKUP(A2,hr_tbl,32,FALSE),IF(N2<=trfc_thresh,VLOOKUP(N2,trk_table,2),CEILING(N2/inputs!\$C\$36,1))))	computes total local tandem trunks required according to total offered load calculation and user-set inputs for maximum trunk occupancy
P	local tdm trk invest per line	=IF(\$C2=0,0,C2*O2/\$AR2*\$AU2)	calculates share of total interoffice facility investment assigned to local tandem trunks
Q	total OS traffic, CCS	=bus_public_lines*bus_OS+res_lines*res_OS	computes total offered load for wire center for operator services traffic
R	OS trks required (equiv per line)	=IF(C2=0,0,1/C2*IF(G2="H",VLOOKUP(A2,hr_tbl,38,FALSE),IF(Q2<=trfc_thresh,VLOOKUP(Q2,trk_table,2),CEILING(Q2/inputs!\$C\$36,1))))	computes total ooperator services trunks required according to total offered load calculation and user-set inputs for maximum trunk occupancy
S	OS trk invest per line	=IF(\$C2=0,0,C2*R2/\$AR2*\$AU2)	calculates share of total interoffice facility investment assigned to operator services trunks
T	tdm invest per line	=IF(C2=0,0,IF(tdm_count>0,('tandem and STP investment'!\$D\$13+inputs!\$C\$37*('wire center investment'!O2+AL2+AF2))*inputs!\$C\$25*(1+intertdm_frac),inputs!\$C\$80))	computes per-line investment in tandem switching equipment, including common equipment and trunk ports; selects surrogate value if company has no tandems in study area
U	tandem wire center inv per line	=IF(tdm_count>0,'tandem and STP investment'!\$D\$15,inputs!\$D\$80)	computes per-line investment in tandem wire center facility; selects surrogate value if company has no tandems in study area
V	OS tdm invest per line	=IF(C2=0,0,IF(OS_tdm_count>0,('tandem and STP investment'!\$H\$15+inputs!\$C\$37*('wire center investment'!R2),inputs!\$C\$81)))	Assumes tandem shares wire center with at least one end office switch
W	OS tandem wire center inv per line	=IF(OS_tdm_count>0,'tandem and STP investment'!\$H\$24,inputs!\$D\$81)	computes per-line investment in operator tandem switching equipment, including common equipment and trunk ports; selects surrogate value if company has no OS tandems in study area
X	operator position inv per line	='tandem and STP investment'!\$H\$19	computes per-line investment in operator tandem wire center facility; selects surrogate value if company has no OS tandems in study area
			repeats investment per line in operator position equipment

Equation Listing

Column	Name	Formula	Description
Y	STP inv per line	=IF(F2=0,0,IF(STP_count>0,(tandem and STP investment!\$D\$23+Z2*(inputs!\$C\$97/2)),inputs!\$C\$79))	computes STP investment per line; if company has no STPs in study area, calculation produces surrogate value
Z	# links required (equiv per line)	=IF(C2=0,0,1/C2*IF(G2="H",VLOOKUP(A2,hr_tbl,43,FALSE),2*CEILING((bus_public_lines*bus_BHCA+res_lines*res_BHCA)*inputs!\$F\$63,1))+IF(OR('loop db inputs'!B2=8,COUNT('loop db inputs'!\$B\$2:\$B\$5000)>50),tandem and STP investment!\$D\$56,0))	total signaling links required by switches in wire center, expressed per line
AA	link investment per line	=IF(\$C2=0,0,C2*Z2/\$AR2*\$AU2+IF(STP_count>0,0,inputs!\$D\$82))	assigns signaling link share of total interoffice facility investment per line; adds surrogate value for tandem A links if company has no STPs in study area
AB	total direct routed access traffic, CCS	=bus_public_lines*bus_acc_dir+res_lines*res_acc_dir	computes total offered load for wire center for direct routed access traffic
AC	total direct routed access trunks (equiv per line)	=IF(C2=0,0,1/C2*IF(G2="H",VLOOKUP(A2,hr_tbl,40,FALSE),IF(AB2<=trfc_thresh,VLOOKUP(AB2,trk_table,2),CEILING(AB2/inputs!\$C\$36,1))+AF2))	computes total direct-routed access trunks required according to total offered load calculation and user-set inputs for maximum trunk occupancy
AD	dedicated access trk inv per line	=IF(\$C2=0,0,C2*AC2/\$AR2*\$AU2)	calculates share of total interoffice facility investment assigned to direct-routed access trunks
AE	total tandem-routed access traffic, CCS	=bus_public_lines*bus_acc_tdm+res_lines*res_acc_tdm	computes total offered load for wire center for tandem routed access traffic
AF	total tandem-routed access trunks (equiv per line)	=IF(C2=0,0,1/C2*IF(G2="H",VLOOKUP(A2,hr_tbl,42,FALSE),IF(AE2<=trfc_thresh,VLOOKUP(AE2,trk_table,2),CEILING(AE2/inputs!\$C\$36,1))))	computes total tandem-routed access trunks required according to total offered load calculation and user-set inputs for maximum trunk occupancy
AG	switched access trk inv per line	=IF(\$C2=0,0,C2*AF2/\$AR2*\$AU2)	calculates share of total interoffice facility investment assigned to switched access trunks
AH	total intraLATA direct-routed traffic, CCS	=(bus_public_lines*bus_LATA_dir+res_lines*res_LATA_dir)*0.5	computes total offered load for wire center for direct-routed intraLATA toll traffic
AI	total intraLATA direct trunks (equiv per line)	=IF(C2=0,0,1/C2*IF(G2="H",VLOOKUP(A2,hr_tbl,34,FALSE),IF(AH2<=trfc_thresh,VLOOKUP(AH2,trk_table,2),CEILING(AH2/inputs!\$C\$36,1))))	computes total direct-routed intraLATA toll trunks required according to total offered load calculation and user-set inputs for maximum trunk occupancy
AJ	intraLATA trk inv (direct) per line	=IF(\$C2=0,0,C2*AI2/\$AR2*\$AU2)	calculates share of total interoffice facility investment assigned to direct-routed intraLATA toll trunks
AK	total intraLATA tandem-routed traffic, CCS	=bus_public_lines*bus_LATA_tdm+res_lines*res_LATA_tdm	computes total offered load for wire center for tandem-routed intraLATA toll traffic

Equation Listing

Column	Name	Formula	Description
AL	total intraLATA tandem trunks (equiv per line)	=IF(C2=0,0,1/C2*IF(G2="H",VLOOKUP(A2,hr_tbl,36,FALSE),IF(AK2<=trfc_thresh,VLOOKUP(AK2,trk_table,2),CEILING(AK2/inputs!\$C\$36,1))))	computes total tandem-routed intraLATA toll trunks required according to total offered load calculation and user-set inputs for maximum trunk occupancy
AM	intraLATA trk inv (tandem) per line	=IF(\$C2=0,0,C2*AL2/\$AR2*\$AU2)	calculates share of total interoffice facility investment assigned to tandem-routed intraLATA toll trunks
AN	total public telephone investment per line	=IF(F2=0,0,'loop db inputs'!G2*inputs!\$C\$120/F2)	calculates investment in public telephone station equipment per line
AO	normalized SA lines	=IF(C2=0,0,1/C2*IF(G2="H",VLOOKUP(A2,hr_tbl,47,FALSE),'loop db inputs'!H2))	calculates special access fraction of total lines; adds host/remote totals for host switches when host/remote calculations enabled
AP	normalized SA investment	=IF(\$C2=0,0,C2*AO2/\$AR2*\$AU2)	assigns special access fraction of interoffice investment per line
AQ	total switched trunks	=IF(C2=0,0,C2*(AF2+'wire center investment'!A12+'wire center investment'!AC2+'wire center investment'!Z2+'wire center investment'!R2+'wire center investment'!O2+'wire center investment'!L2+AL2))	calculates total switched trunks in wire center (not normalized to line count)
AR	total DS-0 equivalents, with SA	=IF(C2=0,0,C2*(AF2+'wire center investment'!A12+'wire center investment'!AC2+'wire center investment'!Z2+'wire center investment'!R2+'wire center investment'!O2+'wire center investment'!L2+AL2+AO2))	calculates total DS-0 circuits required, including special access lines
AS	SA fraction of DS-0s	=1-AQ2/AR2	calculates special access fraction of total DS-0s
AT	total fiber cable investment per line	=IF(C2=0,0,1/C2*(BO2*inputs!\$C\$192))	calculates total optical fiber cable investment per line
AU	total facility investment per line	=AT2+AV2+AW2+AX2+AY2+AZ2+IF(AND(ring_ind=0,loc_tdm_ind=0),CC2,0)	calculates total per-line investment in cable and structure, including poles, manholes, conduit, and buried and underground placement
AV	total aerial structure (poles) inv per line	=IF(C2=0,0,BO2*inputs!\$E\$194/C2)	calculates total investment per line in poles for interoffice facilities
AW	total u/g structure (conduit plcmnt) inv per line	=IF(C2=0,0,BO2*inputs!\$E\$196/C2)	calculates total investment per line in conduit placement for interoffice facilities
AX	total pullbox inv per line	=IF(C2=0,0,BO2*inputs!\$E\$197/C2)	calculates total investment per line in pullboxes for interoffice facilities
AY	total buried plcmnt inv per line	=IF(C2=0,0,BO2*inputs!\$E\$195/C2)	calculates total investment per line in buried placement for interoffice facilities
AZ	total conduit inv per line	=IF(C2=0,0,BO2*inputs!\$E\$198/C2)	calculates total investment per line in conduit for interoffice facilities

Equation Listing

Column	Name	Formula	Description
BA	total DS-1 equivalents (w/sizing factor)	=CEILING(AR2/inputs!\$C\$165/24,1)	calculates total DS-1s for wire center, including facility sizing factor
BB	total DS-3 equivalents	=CEILING(BA2/28,1)	calculates total DS-3 equivalents from DS-1 total
BC	transmission terminal investment per line	=IF(sw_type="R",VLOOKUP(A2,remote_tbl,27,FALSE),IF(C2=0,0,1/C2*(IF(BP2=1,(inputs!\$C\$159-(12-CEILING((BA2+(K2+AI2)/24/inputs!\$C\$165)/7,1)*inputs!\$C\$149)),(IF(BB2<=12,inputs!\$C\$158,inputs!\$C\$157)+IF(CF2>max_rate,(CEILING(CF2/max_rate,1)-1)*inputs!\$C\$157)+CEILING(BB2/3,1)*inputs!\$C\$159))+tandem and STP investment!\$H\$30+inputs!\$C\$164*BB2)+IF(sw_type="H",VLOOKUP(A2,hr_tbl,28,FALSE),0))+1/C2*IF(BO2<inputs!\$C\$163,0,inputs!\$C\$162*(CEILING(BO2/inputs!\$C\$163,1)-1))+IF(AND(ring_ind=0,loc_tdm_ind=0),CB2,0))	computes total transmission terminal investment per line, including regenerators and additional ADMs required for inter-ring connections, tandem and OS tandem ring connections; includes capacity for ring transiting traffic
BD	land investment per line	=IF(F2=0,0,1/F2*VLOOKUP(F2,wc_inv,8))	calculates land investment per line for wire center
BE	total DLC lines	=loop db inputs!I2	repeats total DLC line count for wire center
BF	total common transport inv per line	=P2+AG2+AM2	calculates total investment per line in common (tandem) transport facilities
BG	total dedicated transport per line	=AD2+AP2	calculates total investment in dedicated transport facilities
BH	common fraction	=IF(\$BF2+\$BG2+\$BK2=0,0,\$BF2/(\$BF2+\$BG2+\$BK2))	calculates common transport fraction of total transport facilities investment
BI	direct fraction	=IF(\$BF2+\$BG2+\$BK2=0,0,\$BK2/(\$BF2+\$BG2+\$BK2))	calculates direct transport fraction of total transport facilities investment
BJ	dedicated fraction	=IF(\$BF2+\$BG2+\$BK2=0,0,\$BG2/(\$BF2+\$BG2+\$BK2))	calculates dedicated transport fraction of total transport facilities investment
BK	total direct transport inv per line	=M2+S2+AJ2	calculates total investment per line in direct transport facilities
BL	ring distance	=IF(ring_ind=1,VLOOKUP(A2,ring_list,2,FALSE)*ring_dstnc_adj,0)	obtains ring distance (or spur distance for off-ring wire centers) from distance inputs table for companies for which rings are constructed; distance increased by adjustment factor to account for inter-ring connections
BM	BHCA	=bus_public_lines*bus_BHCA+res_lines*res_BHCA	calculates total busy-hour call attempts for wire center
BN	total BH offered traffic, CCS	=('loop db inputs'!\$E2+'loop db inputs'!\$G2)*inputs!\$D\$73+'loop db inputs'!\$F2*inputs!\$D\$72	calculates total busy-hour offered load for wire center

Equation Listing

Column	Name	Formula	Description
BO	effective interoffice distance	=IF(ring_ind=1,BL2,IF(BT2=0,2*VLOOKUP(A2,dist_tbl,3,FALSE),2*avg_tdm_ds_tnc))	calculates effective interoffice distance as ring distance (or spur distance) if rings are calculated for company, otherwise produces distance to nearest BOC wire center (doubled to allow for route diversity)
BP	small office indicator	=IF('loop db inputs'!D2<sm_off_ind,1,0)	indicates (=1) if switched line total in wire center falls below user-set small office threshold value
BQ	channel bank investment for unmultiplexed SA lines per line	=IF(C2=0,0,1/C2*inputs!\$C\$160*CEILING(inputs!\$C\$161*"loop db inputs"!H2*(1-BE2/C2)/inputs!\$C\$165/24,1))	computes channel bank investment required to multiplex special access lines not served by DLC
BR	spare		
BS	total access circuits	=AR2*BJ2	
BT	ML ind	=IF(ISNA(VLOOKUP(A2,dist_tbl,1,FALSE)),1,0)	missing location indicator; provided as check for wire centers in loop data not appearing in distance data; normally 0
BU	autonomous switch investment per line	=IF(C2=0,0,IF(sw_type="A",1/C2*VLOOKUP(F2/B2/line_fill,sw_inv_tbl,IF(OR(BY2=8,BY2=1),2,8))+VLOOKUP(F2/B2/line_fill,sw_inv_tbl,IF(OR(BY2=8,BY2=1),5,11))-inputs!\$C\$37/6-inputs!\$C\$24*(BE2)/loop db inputs!D2+(Z2*inputs!\$C\$97/2+C2/F2*inputs!\$C\$37*(L2*2+O2+R2+AC2+AF2+AI2*2+AL2)),IF(AND(sw_type="H",B2>1),1/C2*VLOOKUP(F2*(1-1/B2)/B2/line_fill,sw_inv_tbl,IF(OR(BY2=8,BY2=1),2,8))+VLOOKUP(F2*(1-1/B2)/B2/line_fill,sw_inv_tbl,IF(OR(BY2=8,BY2=1),5,11))-inputs!\$C\$37/6-inputs!\$C\$24*(BE2)/loop db inputs!D2+(Z2*inputs!\$C\$97/2+C2/F2*inputs!\$C\$37*(L2*2+O2+R2+AC2+AF2+AI2*2+AL2)),0)))*sw_install_mult	computes investment per line in autonomous, or "stand-alone," switches; if host switch appears in multiple-switch wire centers, autonomous calculation applied to all but the first switch in the wire center
BV	host switch investment per line	=IF(sw_type="H",VLOOKUP(A2,hr_tbl,22,FALSE),0)	obtains host switch investment per line from host/remote calculations
BW	remote switch investment per line	=IF(sw_type="R",VLOOKUP(A2,remote_tbl,21,FALSE),0)	obtains remote switch investment per line from host/remote calculations
BX	aggregate switch investment	=IF(sw_type="",IF(OR(BY2=8,BY2=1),inputs!\$C\$3,inputs!\$C\$2)+inputs!\$C\$4*L_N(F2/B2/inputs!\$C\$18)-inputs!\$C\$37/6-inputs!\$C\$24*(BE2)/loop db inputs!D2+(Z2*inputs!\$C\$97/2+C2/F2*inputs!\$C\$37*(L2*2+O2+R2+AC2+AF2+AI2*2+AL2)),0)/line_fill*sw_install_mult	computes end office switch investment per line when host/remote calculations are not enabled using aggregated investment input values that address host, remote, and autonomous switches
BY	company type	=loop db inputs!B2	repeats operating company type code
BZ	host/remote wire center inv per line	=IF(sw_type="H",VLOOKUP(A2,hr_tbl,20,FA1.SE),IF(sw_type="R",VLOOKUP(A2,remote_tbl,19,FALSE),0))	obtains per-line wire center investment from host/remote calculations

Equation Listing

Column	Name	Formula	Description
CA	spare		determines distance over which facilities must be leased for companies without tandems in study area; the distance is the tandem distance between the BOC wire center to which the switch connects and the BOC tandem
CB	leased facility quasi-investment per line, terminal	=IF(AND(CD2=0,C2>0),term_equiv_inv*AR2/C2,0)	estimates surrogate investment for leased facility terminal equipment using monthly cost factor and representative monthly tariff input
CC	leased facility quasi-investment per line, facility	=IF(AND(CD2=0,C2>0),fac_equiv_inv*AR2/C2,0)	estimates surrogate investment for leased facility using monthly cost factor and representative monthly tariff input
CD	local tandem indicator (1 = yes)	=IF(VLOOKUP(A2,dist_tbl,6,FALSE)=VLOOKUP(A2,dist_tbl,10,FALSE),1,0)	indicates whether company has local tandem
CE	ring indicator (1 = yes)	=IF(OR(ISNA(VLOOKUP(A2,ring_list,1,FALSE)),BT2=1),0,1)	indicates whether rings have been calculated for company
CF	effective DS3s in local ring	=IF(OR(BT2=1,transit_fac=1,CE2=0),0,VLOOKUP(A2,ring_list,5,FALSE)*(1+transit_fac)/2/(1-transit_fac))	obtains total DS-3 count in ring, including contributions from small offices connected by spurs to wire centers on ring; transiting traffic adjustment made in terminal investment calculation
CG	spare		obtains tandem distance for BOC wire center to which small office connects if company has no tandems in study area

Equation Listing

Column	Name	Formula	Description
A	wire center	='wire center investment'!A2	repeats wire center location code
B	total switched lines	='wire center investment'!F2	repeats total switched line count for wire center
C	end office switching inv per line	='wire center investment'!H2	repeats end office switching investment per line
D	MDF/protector inv per line	='wire center investment'!I2	repeats MDF/protector investment per line; applies to non-DLC lines
E	end office wire center inv per line	='wire center investment'!J2	repeats end office wire center investment per line
F	land per line	='wire center investment'!BD2	repeats wire center land investment per line
G	local tdm switching per line	='wire center investment'!T2	repeats tandem investment per line
H	local tdm wire center per line	='wire center investment'!U2	repeats wire center investment for local tandem per line
I	OS tdm switching per line	='wire center investment'!V2	repeats operator tandem investment per line
J	OS tdm wire center per line	='wire center investment'!W2	repeats wire center investment per line for OS tandem
K	OS trunk inv per line	='wire center investment'!S2	repeats investment per line in operator trunks
L	operator position inv per line	='wire center investment'!X2	repeats operator positions investment per line
M	common transport, u/g cable inv per line	=BH2*BV2	assigns cable facility investment to underground common transport
N	common transport, buried cable inv per line	=BI2*BV2	assigns cable facility investment to buried common transport
O	common transport, aerial cable inv per line	=BJ2*BV2	assigns cable facility investment to aerial common transport
P	common transport, pole inv per line	=BK2*BV2	assigns pole investment to common transport including effects of sharing interoffice and feeder structure
Q	common transport, conduit inv per line	=BL2*BV2	assigns conduit investment to common transport including effects of sharing interoffice and feeder structure
R	common transport, pullbox inv per line	=BM2*BV2	assigns pullbox investment to common transport including effects of sharing interoffice and feeder structure
S	common transmission terminal inv per line	=BV2*'wire center investment'!BC2	assigns transmission terminal investment to common transport

Equation Listing

Column	Name	Formula	Description
T	direct transport, u/g cable inv per line	=BH2*BW2	assigns cable facility investment to underground direct transport
U	direct transport, buried cable inv per line	=BI2*BW2	assigns cable facility investment to buried direct transport
V	direct transport, aerial cable inv per line	=BJ2*BW2	assigns cable facility investment to aerial direct transport
W	direct transport, pole inv per line	=BK2*BW2	assigns pole investment to direct transport including effects of sharing interoffice and feeder structure
X	direct transport, conduit inv per line	=BL2*BW2	assigns conduit investment to direct transport including effects of sharing interoffice and feeder structure
Y	direct transport, pullbox inv per line	=BM2*BW2	assigns pullbox investment to direct transport including effects of sharing interoffice and feeder structure
Z	direct transmission terminal inv per line	=BW2*"wire center investment"!BC2	assigns transmission terminal investment to direct transport
AA	dedicated transport, u/g cable inv per line	=BH2*BX2	assigns cable facility investment to underground dedicated transport
AB	dedicated transport, buried cable inv per line	=BI2*BX2	assigns cable facility investment to buried dedicated transport
AC	dedicated transport, aerial cable inv per line	=BJ2*BX2	assigns cable facility investment to aerial dedicated transport
AD	dedicated transport, pole inv per line	=BK2*BX2	assigns pole investment to dedicated transport including effects of sharing interoffice and feeder structure
AE	dedicated transport, conduit inv per line	=BL2*BX2	assigns conduit investment to dedicated transport including effects of sharing interoffice and feeder structure
AF	dedicated transport, pullbox inv per line	=BM2*BX2	assigns pullbox investment to dedicated transport including effects of sharing interoffice and feeder structure
AG	dedicated transmission terminal inv per line	=BX2*"wire center investment"!BC2+"tandem and STP investment"!\$H\$46+"wire center investment"!BQ2	assigns transmission terminal investment to dedicated transport
AH	equiv per line local direct trunks	="wire center investment"!L2	repeats total local direct trunk count expressed per line

Equation Listing

Column	Name	Formula	Description
AI	equiv per line local tandem trunks	='wire center investment'!O2	repeats total local tandem trunk count expressed per line
AJ	equiv per line intraLATA direct trunks	='wire center investment'!AI2	repeats total intraLATA direct trunk count expressed per line
AK	equiv per line intraLATA tandem trunks	='wire center investment'!AL2	repeats total intraLATA tandem trunk count expressed per line
AL	equiv per line direct-routed access trunks	='wire center investment'!AC2+'wire center investment'!AO2	repeats total direct-routed access trunk count expressed per line
AM	equiv per line tandem-routed access trunks	='wire center investment'!AF2	repeats total tandem-routed access trunk count expressed per line
AN	equiv per line operator trunks	='wire center investment'!R2	repeats total operator trunk count expressed per line
AO	SCP inv per line	=IF(OR('loop db inputs'!B2=8,STP_count>0),tandem and STP investment'!\$D\$59,inputs!\$C\$82)	determines SCP investment per line as calculated value or surrogate value for companies without STPs in study area
AP	SCP+STP wire center inv per line	=IF(OR('loop db inputs'!B2=8,STP_count>0),('tandem and STP investment'!\$D\$60+'tandem and STP investment'!\$D\$22),inputs!\$D\$79)	repeats wire center investment per line for SCP and STP
AQ	STP inv per line	='wire center investment'!Y2	repeats STP investment per line
AR	signaling link inv per line	='wire center investment'!AA2	repeats signaling link investment per line
AS	total public telephone inv per line	='wire center investment'!AN2	repeats public telephone station equipment investment per line
AT	total residential annual DEMs per line	=inputs!\$F\$44	repeats average annual residential DEMs per line
AU	total business annual DEMs per line	=inputs!\$F\$43	repeats average annual business DEMs per line
AV	total fdr pullbox inv per line	=IF('loop db inputs'!D2=0,0,'loop db inputs'!O2/'loop db inputs'!D2-BO2*('loop db inputs'!O2/'loop db inputs'!D2)/('loop db inputs'!O2/'loop db inputs'!D2+BG2))	computes feeder pullbox investment per line with effects of sharing with interoffice structure
AW	copper fdr u/g placement per line	=IF('loop db inputs'!D2='loop db inputs'!I2,0,IF('loop db inputs'!N2=0,0,'loop db inputs'!N2/('loop db inputs'!D2-'loop db inputs'!I2)-\$BQ2*('loop db inputs'!N2/('loop db inputs'!D2-'loop db inputs'!I2))/('loop db inputs'!N2/('loop db inputs'!D2-'loop db inputs'!I2)+BD2)))	computes copper feeder underground placement per line with effects of sharing with interoffice structure
AX	fiber fdr u/g placement per line	=IF('loop db inputs'!I2=0,0,IF('loop db inputs'!M2=0,0,'loop db inputs'!M2/'loop db inputs'!I2-\$BQ2*('loop db inputs'!M2/'loop db inputs'!I2)/('loop db inputs'!M2/'loop db inputs'!I2+BD2)))	computes fiber feeder underground placement per line with effects of sharing with interoffice structure

Equation Listing

Column	Name	Formula	Description
AY	copper feeder buried plcmt per line	=IF('loop db inputs'!D2='loop db inputs'!I2,0,IF('loop db inputs'!L2=0,0,'loop db inputs'!L2/('loop db inputs'!D2-'loop db inputs'!I2)-\$BS2*('loop db inputs'!L2/('loop db inputs'!D2-'loop db inputs'!I2)))/('loop db inputs'!L2/('loop db inputs'!D2-'loop db inputs'!I2)+BE2)))	computes copper feeder buried placement investment per line with effects of sharing with interoffice structure
AZ	fiber feeder buried plcmt per line	=IF('loop db inputs'!I2=0,0,IF('loop db inputs'!K2=0,0,'loop db inputs'!K2/('loop db inputs'!I2-\$BS2*('loop db inputs'!K2/('loop db inputs'!I2)))/('loop db inputs'!K2/('loop db inputs'!I2)+BE2)))	computes fiber feeder buried placement investment per line with effects of sharing with interoffice structure
BA	total fdr pole inv per line	=IF('loop db inputs'!D2=0,0,('loop db inputs'!J2/('loop db inputs'!D2-BU2)*('loop db inputs'!J2/('loop db inputs'!D2)+BF2)))	computes feeder pole investment per line with effects of sharing with interoffice structure
BB			
BC			
BD	total transport, u/g plcmt unadj inv per line	='wire center investment'!AW2	repeats underground placement investment for transport facilities for use in feeder sharing calculation
BE	total transport, buried plcmt unadj inv per line	='wire center investment'!AY2	repeats buried placement investment for transport facilities for use in feeder sharing calculation
BF	total transport, pole unadj inv per line	='wire center investment'!AV2	repeats pole placement investment for transport facilities for use in feeder sharing calculation
BG	total transport, pullbox unadj inv per line	='wire center investment'!AX2	repeats pullbox placement investment for transport facilities for use in feeder sharing calculation
BH	total transport, u/g cable inv per line	=IF('loop db inputs'!D2=0,0,'wire center investment'!AT2*inputs!\$C\$172+output!BD2-output!BQ2*BD2/(BD2+('loop db inputs'!M2+'loop db inputs'!N2)/('loop db inputs'!D2)))	computes final total transport underground investment per line including effects of structure sharing with feeder
BI	total transport, buried cable inv per line	=IF('loop db inputs'!D2=0,0,'wire center investment'!AT2+'wire center investment'!BO2*inputs!\$C\$171)*inputs!\$C\$169+BE2-BS2*BE2/(BE2+('loop db inputs'!K2+'loop db inputs'!L2)/('loop db inputs'!D2)))	computes final total transport buried investment per line including effects of structure sharing with feeder
BJ	total transport, aerial cable inv per line	='wire center investment'!AT2*inputs!\$C\$178	calculates total aerial cable investment per line for transport
BK	total transport, pole inv per line	=IF('loop db inputs'!D2=0,0,BF2-BU2*BF2/(BF2+('loop db inputs'!J2/('loop db inputs'!D2)))	computes final total transport pole investment per line including effects of structure sharing with feeder
BL	total transport, conduit inv per line	='wire center investment'!AZ2	computes final transport conduit investment per line
BM	total transport, pullbox inv per line	=IF('loop db inputs'!D2=0,0,BG2-BO2*BG2/(BG2+('loop db inputs'!O2/('loop db inputs'!D2)))	computes final total transport pullbox investment per line including effects of structure sharing with feeder

Equation Listing

Column	Name	Formula	Description
BN	min pullbox inv per line (i/o, fdr)	=IF('loop db inputs'!\$D2=0,0,MIN('loop db inputs'!\$O2/'loop db inputs'!\$D2,BG2))	determines minimum pullbox investment per line between interoffice and feeder facilities for use in structure sharing calculation
BO	basic pullbox inv reduction per line	=BN2*inputs!\$C\$184	computes pullbox investment reduction per line resulting from sharing of structure between interoffice and feeder facilities; applied to both interoffice and feeder totals
BP	min u/g plcmt inv per line (i/o, fdr)	=IF('loop db inputs'!\$D2=0,0,MIN(BD2,('loop db inputs'!\$M2+'loop db inputs'!\$N2)>'loop db inputs'!\$D2))	determines minimum underground placement investment per line between interoffice and feeder facilities for use in structure sharing calculation
BQ	basic u/g plcmt reduction per line	=BP2*inputs!\$C\$184	computes underground placement investment reduction per line resulting from sharing of structure between interoffice and feeder facilities; applied to both interoffice and feeder totals
BR	min buried plcmt inv per line (i/o, fdr)	=IF('loop db inputs'!\$D2=0,0,MIN(BE2,('loop db inputs'!\$K2+'loop db inputs'!\$L2)>'loop db inputs'!\$D2))	determines minimum buried placement investment per line between interoffice and feeder facilities for use in structure sharing calculation
BS	basic buried plcmt reduction per line	=BR2*inputs!\$C\$184	computes buried placement investment reduction per line resulting from sharing of structure between interoffice and feeder facilities; applied to both interoffice and feeder totals
BT	min pole inv per line (i/o, fdr)	=IF('loop db inputs'!\$D2=0,0,MIN(BF2,'loop db inputs'!\$J2/'loop db inputs'!\$D2))	determines minimum pole investment per line between interoffice and feeder facilities for use in structure sharing calculation
BU	basic pole reduction per line	=BT2*inputs!\$C\$184	computes pole investment reduction per line resulting from sharing of structure between interoffice and feeder facilities; applied to both interoffice and feeder totals
BV	common fraction	='wire center investment'!BH2	repeats common investment fraction of total for use in assigning various investment to common transport
BW	direct fraction	='wire center investment'!B12	repeats direct investment fraction of total for use in assigning various investment to direct transport
BX	dedicated fraction	='wire center investment'!BJ2	repeats dedicated investment fraction of total for use in assigning various investment to dedicated transport

Workbook: R50A_switching_i.xls
Worksheet: output

Equation Listing

HAI Model, v5.0A
Switching/Interoffice Module

Column	Name	Formula	Description
BY	ML indicator	='wire center investment'!B12	repeats missing location indicator; normally zero